

SAS® GLOBAL FORUM 2018

USERS PROGRAM

Outputting Your Data to Microsoft Excel Is Inevitable. So Is the SAS®
ODS Excel Destination.

Claudia Hauck & Hezekiah Bunde

Educational Testing Service

April 8 - 11 | Denver, CO

#SASGF

Outputting Your Data to Microsoft Excel Is Inevitable. So Is the SAS® ODS Excel Destination.

Claudia Hauck & Hezekiah Bunde
Educational Testing Service

ABSTRACT

Microsoft Excel is one of the most used tools wherever data is used, stored, or analyzed. Many SAS® users often resort to producing Excel output and working outside of the SAS® environment to finesse the deliverables...until now. With the SAS® Output Delivery System (ODS) Excel destination, data manipulation and visualization is now possible. Native Excel files and graphs can now be created and customized.

It was just a matter of time before this magnificent tool became a reality. ODS Excel is here to stay! Novice programmers with little or no experience at all with ODS output to experienced professionals will instantly experience its benefits. This e-poster demonstrates with easy-to-follow steps how to deliver your data from SAS to Excel. Users will realize time-saving benefits by pre-defining their preferences and avoid performing manual and repetitive tasks such as creating multiple sheets, adding color, titles, graphs, headers, footers, and so on.

INTRODUCTION

The ubiquitous nature of Microsoft Excel makes it inevitable as a medium of data. While many programmers still produce Excel files and work on finessing the output in Excel, they are not benefitting from the advent of SAS® ODS Excel which allows many of these manipulations within the SAS® environment. Users can now save time by pre-defining their preferences and avoid performing manual and repetitive tasks such as creating multiple sheets, adding color, titles, graphs, headers, footers, etc.

In this e-poster presentation, three tables and two graphs are created to demonstrate ODS Excel capabilities using two SAS® datasets: *shoes* and *stocks*. With the former dataset, a chart and three Excel tables are created using *proc report* and *proc tabulate*. The latter dataset is used to create a graph using *proc sgplot*. All output is rendered in native Excel files! This e-poster exhibits the effortless execution of customized output including (but not limited to) traffic lighting, formatting and graphical display.

Outputting Your Data to Microsoft Excel Is Inevitable. So Is the SAS® ODS Excel Destination.

Claudia Hauck & Hezekiah Bunde
Educational Testing Service

METHODS

SAS® datasets were used to demonstrate the ease and convenience of creating Excel graphs and tables.

- **SAS® ODS Excel table with options** – The sashelp.shoes dataset is used to demonstrate the ease and availability of multiple options at the user's disposal

```
ods excel file="C:\Table1.xlsx";
title1          j=c "Have SAS® work for YOU !";
title2          j=c "What a great title for my report!";
ods excel options (embedded_titles = "yes" sheet_name = "#byvall");
proc report data = sashelp.shoes style(header)=[background=silver bordercolor=black borderwidth=2];
  by Region;
  column Subsidiary Sales Inventory Returns;
  define Sales/display "Sales";
  compute Sales;
  if Sales ne . then do;
  if Sales < 20000 then call define (_row_, "style", "style={background=light red}");
  if Sales >1000000 then call define (_row_, "style", "style={background=light green}");
  end;
  endcomp;
  define Inventory/style(header)=[background=lightblue]"Inventory";
run;
ods excel close;

title;
ods excel file="C:\Table2.xlsx";
proc format;
  value temp 60000-70000=yellow;
run;
ods excel options(sheet_name='Sales' autofilter = '2-3' embedded_titles='on');
title3 j=c 'Highlight Cells Rule and Filtering';
proc tabulate data=sashelp.shoes;
  var Sales;
  class Region Product Subsidiary;
  table Region*(Subsidiary*Product all=[label='Region Total'])
  all= [Label='Grand Total'],Sales=[Label='Total Sales']*f=dollar12.*sum=[label=' ']*[Style=[background =temp.]]/
  box='World Domestic Sales';
run;
ods excel close;

title;
ods excel file= "C:\Table3.xlsx" style=ocean;
ods excel options(embedded_titles='on' sheet_name="Cairo" tab_color="green");
title4 j=c 'Excel Worksheet with style option';
proc report data=sashelp.shoes (where=(Subsidiary="Cairo"));
  columns Product Sales Inventory;
run;
ods excel close;
```

- **Excel printing options**– A set-up of the print options is shown.

Action	Default	Options
embedded_titles	no	'off', 'on'
center_Horizontal	off	'off', 'on'
center_vertical	off	'off', 'on'
print_header_margin	0.5	'number'
print_footer_margin	0.5	'number'
pages_fitwidth	None	'number'
pages_fitheight	None	'number'
hidden_columns	None	'number-list-range'
hidden_rows	None	'number-list-range'
print_area	None	'item'
orientation	Portrait	'Portrait', 'Landscape'
row_repeat	None	'None', 'Header', 'number', 'number-range'
zoom	100	'number'
print_footer		'text-string'
autofilter	None	'None', 'All', 'range'

Center on page
 Horizontally
 Vertically

Scaling
 Adjust to: 100 % normal size
 Fit to: 1 page(s) wide by 1 tall

Orientation
 Portrait Landscape

Scaling
 Adjust to: 100 % normal size
 Fit to: 1 page(s) wide by 2 tall

- **Charts/Graphs** – Stock chart and Graph created using *stocks* and *shoes* datasets.

```
title;
ods excel file="C:\Stock Graph.xlsx";
title 'Stock Trends in USA';
proc sgplot data=sashelp.stocks;
  styleattrs backcolor=lightgray;
  series x=date y=close/group=stock;
run;
ods excel close;
```

```
title;
ods excel file= "C:\PieChart.xlsx";
title "Shoe Sales in Africa by Subsidiary";
goptions colors=cback=LightGrey;
proc gchart data=sashelp.shoes (where=(Region="Africa"));
  pie Subsidiary/ sumvar=sales other=0
  clockwise value=none slice=outside
  percent=outside noheading;
run;
ods excel close;
```

Outputting Your Data to Microsoft Excel Is Inevitable. So Is the SAS® ODS Excel Destination.

Claudia Hauck & Hezekiah Bunde
Educational Testing Service

RESULTS




Chart 1 : Excel Stock Chart




Chart 2 : Excel Pie Chart

Outputting Your Data to Microsoft Excel Is Inevitable. So Is the SAS® ODS Excel Destination.

Claudia Hauck & Hezekiah Bunde
Educational Testing Service

RESULTS cont.

Have SAS work for YOU! What a great title for my report			
Region=Middle East			
Subsidiary	Sales	Inventory	Total Returns
Al-Khobar	\$15,062	\$44,658	\$765
Al-Khobar	\$340,201	\$616,745	\$13,251
Al-Khobar	\$261,445	\$711,246	\$6,979
Al-Khobar	\$1,380	\$13,343	\$62
Al-Khobar	\$66,796	\$201,779	\$3,052
Al-Khobar	\$226,514	\$447,377	\$5,966
Al-Khobar	\$241,820	\$611,686	\$6,167
Dubai	\$90,972	\$403,259	\$4,049
Dubai	\$419,336	\$868,115	\$17,552
Dubai	\$324,312	\$746,970	\$8,159
Dubai	\$17,492	\$59,985	\$734
Dubai	\$387,680	\$1,238,698	\$12,768
Dubai	\$1,487	\$11,432	\$70
Dubai	\$233,374	\$416,012	\$8,102
Dubai	\$435,891	\$1,292,400	\$12,999
Tel Aviv	\$65,248	\$222,165	\$2,320
Tel Aviv	\$1,298,717	\$2,881,005	\$57,362
Tel Aviv	\$253,814	\$631,479	\$8,727

Table 1 : Traffic-lighting via PROC REPORT

Highlight Cells Rule and Filtering				
World Domestic Sales			Total Sales	
Region	Subsidiary	Product		
Africa	Addis Ababa	Boot	\$29,761	
		Men's Casual	\$67,242	
		Men's Dress	\$76,793	
		Sandal	\$62,819	
		Slipper	\$68,641	
		Sport Shoe	\$1,690	
		Women's Casual	\$51,541	
		Women's Dress	\$108,942	
		Algiers	Boot	\$21,297
			Men's Casual	\$63,206
	Men's Dress		\$123,743	
	Sandal		\$29,198	
	Slipper	\$64,891		
	Sport Shoe	\$2,617		
Women's Dress	\$90,648			

Table 2 : Highlighting and filtering via PROC TABULATE

Excel Worksheet with style option		
Product	Total Sales	Total Inventory
Boot	\$4,846	\$18,965
Men's Casual	\$360,209	\$1,063,251
Men's Dress	\$4,051	\$45,962
Sandal	\$10,532	\$50,430
Slipper	\$13,732	\$54,117
Sport Shoe	\$2,259	\$20,815
Women's Casual	\$328,474	\$940,851
Women's Dress	\$14,095	\$51,145

Table 3 : Excel sheet with style option

Outputting Your Data to Microsoft Excel Is Inevitable. So Is the SAS® ODS Excel Destination.

Claudia Hauck & Hezekiah Bunde
Educational Testing Service

CONCLUSION

ODS Excel is an immensely significant SAS® addition. ODS Excel offers an effortless way to transfer your SAS® output to native excel files. ODS Excel allows the user complete control of the Excel destination while still in the SAS® environment. The user is encouraged to explore ODS Excel options that include (but are not limited to) those illustrated here. Doing so will afford newfound ways of managing and visualizing data.

REFERENCES

- Zender, Cynthia L., 2016. “That is all Right: More Complex Reports”. Proceeding of the SAS Global Forum 2016 Conference.
Available at: <http://support.sas.com/resources/papers/proceedings16/SAS5762-2016.pdf>
- Parker, Chevell, 2016. “A Ringside Seat: The ODS Excel Destination versus the ODS ExcelXP Tagset”. Proceedings of the SAS Global Forum 2016 Conference.
Available at: <https://support.sas.com/resources/papers/proceedings16/SAS5642-2016.pdf>
- Parker, Chevell 2015. “Secrets from a SAS Technical Support Guy: Combining the Power of the SAS Output Delivery System with Microsoft Excel Worksheets” Proceedings of the SAS Global Forum 2014 Conference.
Available at: <http://support.sas.com/resources/papers/proceedings14/SAS177-2014.pdf>
- Huff, Gina 2016. “An ‘Excel’lent Journey: Exploring the New ODS EXCEL Statement” Proceedings of the SAS Global Form 2016 Conference.
Available at <http://support.sas.com/resources/papers/proceedings16/2780-2016.pdf>
- SAS Documentation. “SAS 9.4 Output Delivery System: User’s guide, Fifth Edition” 1994-2012. Pages 292-320.
Available at <http://documentation.sas.com/api/docsets/odsug/9.4/content/odsug.pdf?locale=en#nameddest=bookinfo>

CONTACT INFORMATION

Claudia Hauck
Psychometric Analysis & Research (PAR) | ETS | 660 Rosedale Road, Anrig Hall P253 |
Princeton, NJ 08541 | Phone: 609-734-1099 | chauck@ets.org

Hezekiah Bunde
Data Analysis & Research Technologies (DART) | ETS | 660 Rosedale Road, Thurstone Hall T234 |
Princeton, NJ 08541 | Phone: 609-734-1569 | hbunde@ets.org

Outputting Your Data to Microsoft Excel Is Inevitable. So Is the SAS® ODS Excel Destination.

Claudia Hauck & Hezekiah Bunde

Educational Testing Services (ETS)

Sample Code

```
*-----;  
* Traffic Lighting in Excel ;  
*-----;  
  
ods excel file="C:\Table1.xlsx";  
title1          j=c "Have SAS® work for YOU !";  
title2          j=c "What a great title for my report!";  
ods excel options (embedded_titles = "Yes" sheet_name = "#byval1");  
proc report data = sashelp.shoes style(header)=[background=silver bordercolor=black borderwidth=2];  
  by Region;  
  column Subsidiary Sales Inventory Returns;  
  define Sales/display "Sales";  
  compute Sales;  
  if Sales ne . then do;  
  if Sales < 20000 then call define(_row_, "style", "style={background=light red}");  
  if Sales >1000000 then call define(_row_, "style", "style={background=light_green}");  
  end;  
  endcomp;  
  define Inventory/style(header)=[background=lightblue]"Inventory";  
run;  
ods excel close;
```

```

*-----;
* Highlighting Cells Rules & Filtering ;
*-----;
title;
ods excel file="C:\Table2.xlsx";
proc format;
    value temp 60000-70000=yellow;
run;

ods excel options(sheet_name='Sales' autofilter = '2-3' embedded_titles='on');
title3 j=c 'Highlight Cells Rule and Filtering';
proc tabulate data=sashelp.shoes;
    var Sales;
    class Region Product Subsidiary;
    table Region*(Subsidiary*Product all=[label='Region Total'])all= [Label='Grand Total'],
    Sales=[Label='Total Sales']*f=dollar12.*sum=[label=' ']*[Style=[background =temp.]]/box='World Domestic Sales';
run;
ods excel close;

*-----;
* Worksheet style option ;
*-----;

title;
ods excel file= "C:\Table3.xlsx" style=ocean;
ods excel options(embedded_titles='on' sheet_name="Cairo" tab_color="green");
title4 j=c 'Excel Worksheet with style option';
proc report data=sashelp.shoes (where=(Subsidiary="Cairo"));
    columns Product Sales Inventory;
run;
ods excel close;

```



```

*-----;
* Piechart ;
*-----;

title;
ods excel file= "C:PieChart.xlsx";
title5 "Shoe Sales in Africa by Subsidiary";
goptions colors=cback=LightGrey;
proc gchart data=sashelp.shoes (where=(Region="Africa"));
    pie Subsidiary/ sumvar=sales other=0 clockwise value=none slice=outside percent=outside noheading;
run;
ods excel close;

*-----;
* Stock Chart ;
*-----;

title;
ods excel file="C:\Stock_Graph.xlsx";
title6 'Stock Trends in USA';
proc sgplot data=sashelp.stocks;
    styleattrs backcolor=lightgray;
    series x=date y=close/group=stock;
run;
ods excel close;

*-----;
* Print options ;
*-----;
*adding printing options to above tables code - optional;
ods excel options (
center_horizontal      = "on"
center_vertical        = "off"
pages_fitwidth         = "1"
pages_fitheight        = "2"
print_header_margin    = "1"
print_footer_margin    = "1"
hidden_columns         = "4"
hidden_rows            = "12"
orientation            = "Portrait"
row_repeat             = "Header" /* and/or column_repeat=*/
zoom                   = "90"
print_footer           = "So ready to get printed");

```